# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=hspec-megaparsec
pkgname=haskell-hspec-megaparsec
pkgver=2.2.0
pkgrel=166
pkgdesc="Utility functions for testing Megaparsec parsers with Hspec"
url="https://github.com/goldfirere/hspec-megaparsec"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-hspec-expectations' 'haskell-megaparsec')
makedepends=('ghc' 'haskell-hspec')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('347e1d5275fbd52bc0a51a1106aab8de123034ec34d5f0a113af76136009bc16d2c570aa747e903e050899cf3d53a091e765af5aa67dd7eb3ee7aef6e62054a4')

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE.md "$pkgdir"/usr/share/licenses/$pkgname/LICENSE.md
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE.md
}
